

0.S2 ジュニア版 特別章(2) 積み上げ式 Procedure プログラミング

GAUSSに限らず、プログラミング方法には、そのプログラミング言語のアーキテクチャに反さないかぎり、それぞれの考え方がありまたそれに対応した方法があります。ここでは、コンパイルしないことを前提に、論文やレポート等の付属プログラムとして、誰にでも一目でプログラムの構造が理解できる判読可能でかつ単一ファイルにおさまるプログラミング法を提案します。また、私の著作のそのほとんどがこの体裁を採用しています。GAUSSにはDOSの時代を経て発展してきていますので、一部研究者には、set ファイルを用いたり、複数ファイルをCライクにinclude 命令でもってつなぎ合わせたりして実行させるものもあります。そうした方法も、ある意味で正統プログラミング法なのですが、ここでは(1)データファイルを除きプログラム部分は単一ファイル(2)procedureを細かく積み上げて内部組込み関数と同じ扱いをさせるようにさせる(3)速さや容量よりも容易に読むことができることを重んじる、そういったプログラミング方法を提案します。もちろん、この方法はLight版においても共通に用いられるものです。

単一 Procedure の扱い方

これまでも学習してきましたように、procedureは1つのある複雑な関数を計算する関数部分です。基本的にFortranなどのsubroutineとは扱いが多少異なりますので、ここではあえてそうは呼ばないことにします。2つのサイコロをn回振るプログラムをします。

プログラム

```
new;  
cls;  
print dice2(5);  
  
proc dice2(n);  
    local i,m,p1,p2;  
    m=zeros(n,2);  
    i=1;  
    do while i<=n;  
        p1=ceil(6*randu(1,1));  
        p2=ceil(6*randu(1,1));  
        m[i,]=p1~p2;  
        i=i+1;  
    endo;  
    retp(m);  
endp;
```

画面表示

6.0000000	3.0000000
4.0000000	3.0000000
6.0000000	2.0000000
5.0000000	6.0000000
6.0000000	4.0000000

プログラム各行の説明

- 1 行目 メモリを初期化する。
- 2 行目 スクリーンをクリアする。
- 3 行目 関数 dice2 の中にインプットとして 5 を代入して、その結果を画面表示させる。
- 4 行目 単に区別をする改行。複数行改行してもかまわない。
- 5 行目 ~ 16 行目 インプットが n、アウトプットが m の dice2 という procedure の塊。
- 6 行目 この procedure 内だけで使われるローカル変数は、i, m, p1, p2 の 4 つ。これらはたとえ procedure 外部で別に定義されていても影響を及ぼさない。
- 7 行目 行列変数 m を n 行 2 列の零行列として確保しておく。以降行なうような index を用いた代入で何も無い所に数字を代入することはできない。この作業は必須。
- 8 行目 変数 i の初期値に 1 を設定する。
- 9 行目 i が n 以下の間、endo までの部分の Do ループを続ける。送でなければ、endo の後に行く。
- 10 行目 組込み関数 rndu を用いて、一様乱数を 1 行 1 列分（つまり 1 個）発生させて、それを 6 倍したものを組込み関数 ceil を用いて小数点以下切り上げをして、それを変数 p1 に代入する。つまり、{1, 2, 3, 4, 5, 6} の中から 1 つ引くこと。
- 11 行目 同様にして {1, 2, 3, 4, 5, 6} の中から 1 つ引いて、結果を変数 p2 に代入する。
- 12 行目 変数 p1 と p2 を水平方向にマージしたものを行列変数 m の i 行目に格納する。
- 13 行目 i を 1 つ増やして、それを新たに i として Do に戻る。
- 14 行目 Do ループの終わりの部分を表す。
- 15 行目 リターンとして、行列変数 m を外部に返す。
- 16 行目 procedure の終わりの部分を表す。GAUSS では、これは何時でも必要。

注意事項

- 1) GAUSS においてはべた書きをしない。見易いように字下げをする。
- 2) 関数 Procedure は通常、下部に置かれる。改行によって独立させて見やすくする。
- 3) Procedure 内部では対応する動作の始めと終わりは同じレベルの字下げにする。

以上のプログラムは、後半の自作関数 procedure 部分と前半のそれを呼び出す部分の 2 つからなっています。すなわち、インプット n が与えられているとき、1 から n 回まで 2 つの

乱数からサイコロの目を出し、それをいったん p1 と p2 に入れておいて、それらを水平方向に並べたものを、あらかじめ $n \times 2$ の零行列で確保されている m の i 行目に置くというものです。それを 1 から n 行まで繰り返して、i が n 以下でなくなれば Do ~ endo のループを脱出します。そういう意味で、do と endo のところは字下げがそろっていなければ見通しが悪くなります。最後に、n 行までサイコロの目が 2 列ずつ入った行列変数 m をこの関数 procedure の外部に返しているわけです。この proc がら endp まだが 1 つの自作関数であって、一度そのファイル上で作成されれば、そのファイル上にその関数があるかぎり、通常の GAUSS 標準装備の組み込み関数とまったく同等に使用できます。イメージとして、標準装備の組み込み関数(ここでは rndu や ceil など)がさらにそのファイルの奥に存在すると言うか、あたかもその最下部で機能しているとでも想像してみてください。この自作関数 dice2 のインプットに 5 を入れたものを冒頭部分で使って、その結果を画面表示させているのです。

プログラム

```
new;  
cls;  
call dice2(5);  
  
proc dice2(n);  
    local i,m,p1,p2;  
    m=zeros(n,2);  
    i=1;  
    do while i<=n;  
        p1=ceil(6*rndu(1,1));  
        p2=ceil(6*rndu(1,1));  
        m[i,]=p1~p2;  
        i=i+1;  
    endo;  
    print/rz m;  
    retp(m);  
endp;
```

なお、上のように procedure 内部に print 文があつて m を出力をし、かつリターンとして m が存在する場合、冒頭では call 文を使うことによって、その関数だけを機能させてリターンを止めてしまうことも可能です。また、もともと **proc(0)=dice2(n);** と冒頭を改造し、retp 文をなくすことによって、リターンなしの procedure にすることも可能です。

複数の procedure の扱い方（その1 組込み関数と同等に）

まずは、自作の関数を組込み関数と同等に用いて、さらに別の procedure の中で使う例を見てみましょう。以下では、2つのサイコロをn回振って、その目の合計がsum以下である確率を求めるシミュレーションをしてみます。

プログラム

```
new;
cls;
print "Prob=" simdice(5,8);

proc simdice(n,sum);
    local m,p;
    m=(sumc(dice2(n)').<=sum);
    p=sumc(m)/n;
    retp(p);
endp;

proc dice2(n);
    local i,m,p1,p2;
    m=zeros(n,2);
    i=1;
    do while i<=n;
        p1=ceil(6*randu(1,1));
        p2=ceil(6*randu(1,1));
        m[i,]=p1~p2;
        i=i+1;
    endo;
    print/rz m;
    retp(m);
endp;
```

画面表示

```
Prob=

          3          3
          5          1
          6          6
          3          4
```

0.80000000

プログラム各行の説明

- 1 行目 メモリを初期化する。
- 2 行目 スクリーンをクリアする。
- 3 行目 関数 simdice の中にインプットとして n と sum に対してそれぞれ 5 と 8 をを代入して、その結果を、Prob=というコメントとともに、画面表示させる。
- 4 行目 単に区別をする改行。複数行改行してもかまわない。
- 5 行目～10 行目 インプットが n と sum、アウトプットが p の simdice という procedure の塊。
- 6 行目 この procedure 内だけで使われるローカル変数は、m,p の 2 つ。これらは、たとえば procedure 外部で別に定義されていても影響を及ぼさない。実際、下の dice2 で違った意味で定義されて計算に使われている。
- 7 行目 このファイル上にある自作関数 dice2(n)を用い、それを転置させたものの列ごとの和を組み込み関数 sumc で求め(つまり行ごとの和を求め)それが要素ごとの比較で(ドットは要素ごとを表す) sum 以下であるものを 1、そうでないものを 0 とする、合計 $n \times 1$ の 0 と 1 の結果ベクトルを変数 m に代入する。
- 8 行目 変数 m の合計を求め、それを n で割ることにより、これを変数 p に代入する。
- 9 行目 この procedure のアウトプットとして p を外部に返す。
- 10 行目 この procedure の終わりの部分を表す。
- (以降) 前述の dice2 の procedure と同じ。dice2 は組み込み関数と同等に機能する。

上では、dice2 という下部にある procedure があたかも GAUSS 標準装備の組み込み関数と同等に機能するかのごとく用いて、また別の simdice という procedure の中で用いられています。ここでは、最下部に置かれた dice2 という procedure のリターン m を、その上の simdice という別の procedure のインプットとする必要は必ずしもありません。そのファイル上に、そういう名前の自作関数があれば、その関数も組み込み関数と同様に自由に使えるわけです。このようにすることによって、ここでは、2 つのサイコロを n 回振る関数と 2 つのサイコロを n 回振って出た目の合計が sum 以下になる確率を求める関数との 2 つに作業を分割できるのです。こうした方がプログラムの読み手にとって、よりわかりやすい論理構造を示すことができます。我々人間は、ある一定の長さや論理展開をする他人のプログラムを理解するには相当の根気が必要です。これをそれぞれの作業を短くすることによって、誰にでも理解できるシンプルな論理展開に分割して、それを細かく積み上げていくのです。ただし、ここでは上部の simdice は下部の dice2 という procedure を計算するにあたって必要とします。GAUSS と言えども、バージョンを重ねるごとに新しい関数ができるわけで、このことはさほど問題とは思われません。また、そのバージョンに最新バージョン

の関数がなければ、それに相当するものを自作すればよいわけです。

なお、画面結果では、実際のサイコロの目が表示されてしまっていますが、これは下部の dice2 という procedure の内部に pprint 文が含まれているからです。確率だけを表示したいのなら、この print 文のところを 1 行消去するか、またはこの procedure をこのまま残したいのなら、上部の simdice の procedure の dice2 を用いる前後を、

```
proc simdice(n,sum);  
    local m,p;  
    screen off;  
    m=(sumc(dice2(n')).<=sum);  
    screen on;  
    p=sumc(m)/n;  
    retp(p);  
endp;
```

とすることによって、その部分の画面表示（スクリーン）を OFF にして止めることも可能です。ON にすればそこからまた通常通り画面表示されます。ただし、m には計算結果が入っているわけですから、その後の結果がなくなってしまうわけではありません。

注意事項

- 1) 自作 procedure はファイル上のどこにおいても組込み関数と同等に使える。
- 2) 自作 procedure を別の procedure で使う場合、必ずしもインプットとして扱わなくてもかまわない。
- 3) 組込み関数は最深部（最下部）にあるかのごとく想像して、より原始的な自作関数を下部から上部へと積み上げていくプログラミング方法を推奨する。

複数の procedure の扱い方（その2 片方をもう片方のインプットとして扱う）

今度は、同じことをするプログラムを片方の procedure をもう片方のインプットとして扱う方法を見てみましょう。

プログラム

```
new;  
cls;  
print "Prob=" simdice(dice2(5),8);  
  
proc simdice(m,sum);  
    local x,p;  
    x=(sumc(m').<=sum);
```

```

        p=sumc(x)/rows(x);
        retp(p);
    endp;

proc dice2(n);
    local i,m,p1,p2;
    m=zeros(n,2);
    i=1;
    do while i<=n;
        p1=ceil(6*randu(1,1));
        p2=ceil(6*randu(1,1));
        m[i,]=p1~p2;
        i=i+1;
    endo;
    print/rz m;
    retp(m);
endp;

```

プログラム各行の説明

- 1 行目 メモリを初期化する。
 - 2 行目 スクリーンをクリアする。
 - 3 行目 関数 simdice の中にインプットとして m と sum に対してそれぞれ dice2(5) と 8 を代入して、その結果を、Prob=というコメントとともに、画面表示させる。インプット m の方は、関数 dice2(n) の入れ子になっている。
 - 4 行目 単に区別をする改行。複数行改行してもかまわない。
 - 5 行目 ~ 10 行目 インプットが m と sum、アウトプットが p の simdice という procedure
 - 6 行目 この procedure 内だけで使われるローカル変数は、x, p の 2 つ。
 - 7 行目 インプット m を転置させたものの列ごとの和を組み込み関数 sumc で求め（つまり行ごとの和を求め）それが要素ごとの比較で（ドットは要素ごとを表す）sum 以下であるものを 1、そうでないものを 0 とする、合計 $n \times 1$ の 0 と 1 の結果ベクトルを変数 x に代入する。
 - 8 行目 変数 x の合計を求め、それを x の行数で割ることにより、これを変数 p に代入。
 - 9 行目 この procedure のアウトプットとして p を外部に返す。
 - 10 行目 この procedure の終わりの部分を表す。
- (以降) 前述の dice2 の procedure と同じ。dice2 は組み込み関数と同等に機能する。

注意事項

1) procedure を呼び出す際、丸括弧の中には別の関数も入れ子で書ける。
2) 丸括弧の関数の入れ子が可能なのは呼び出し部分であって procedure の冒頭でない。
ほとんど同じ動作をするプログラムですが、プログラム言語によっては文法制約上または慣習上、この方法しか用いることのできないケースもありますから、そういう場合には、別の procedure のアウトプットリターンを、また別の procedure のインプットとして用いていくことを繰り返すことになります。GAUSS では、この方法に固執することは全く必要ありません。アルゴリズム上すっきりとした方、または、自分の説明しようとしていることにより近い論理構造の方法を選ぶべきでしょう。

複数の procedure の扱い方 (その3 関数自体を別の procedure で最適化する方法)

今度は、すこし違うことをやってみます。ある関数の最小値または最大値、あるいはその関数をゼロと置いた時の方程式の解を別の procedure で見つけ出す方法です。これには、呼び出す際に特別なやり方があります。

プログラム

```
new;  
cls;  
call gridsolve(&f,0,1,0.001);  
  
proc gridsolve(&f,start,finish,step);  
    local x,x0,f;proc;  
    x=seqa(start,step,(finish-start)/step+1);  
    x0=x[minindc(abs(f(x)))];  
    print "Solution: x0=" x0;  
    retp(x0);  
endp;  
  
fn f(x)=x^3+2*x-1;
```

画面表示

```
Solution: x0=      0.45300000
```

プログラム各行の説明

- 1 行目 メモリを初期化する。
- 2 行目 スクリーンをクリアする。
- 3 行目 関数 gridsolve の中にインプットとして関数としての f、それに通常のインプット変数として、start に 0 を finish に 1 をそして step に 0.001 を入れ計算。

- 4 行目 単に区別をする改行。複数行改行してもかまわない。
- 5 行目 ~ 11 行目 インプットが関数 f とそれに通常の変数 $start, finish, step$ の計 4 つ。アウトプットが $x0$ の `girdsolve` という procedure の塊。
- 6 行目 この procedure 内だけで使われるローカル変数は、 $x, x0$ の 2 つに加えて、 f という proc 分類の関数。
- 7 行目 組込み関数 `seqa` を用いて、 $start$ から $step$ ごとに $(finish-start)/step+1$ 個の数列を作成し x に代入する。
- 8 行目 その x を関数 f に入れた結果を組込み関数 `abs` で絶対値を取ったもの、さらに組込み関数 `minindc` で最小値を与えるインデックス番号（上から何番目かという行数）を得て、それを最終的に変数 x の四角括弧の中に入れて x のそのインデックス番号目の値を $x0$ に代入する。
- 9 行目 変数 $x0$ を、`Solution: $x0=$` というコメントとともに、画面表示する。
- 10 行目 この procedure のリターンとして $x0$ を外部に返す。ただし、冒頭で `call` 文でこの関数を呼び出しているので、結果が 2 重になることはない。Print 文で呼び出すと、すでにこの関数の中に `print` 文があるので 2 重になる。
- 11 行目 この procedure の終わりの部分を表す。
- 12 行目 単に区別をする改行。複数行改行してもかまわない。
- 13 行目 1 行定義命令 `fn` を用いて関数 $f(x) = x^3 + 2x - 1$ と定義する。これは、
- ```
proc f(x);
 retp(x^3+2*x-1);
endp;
```
- としても同じことである。 `f n` は procedure の特殊形に分類される。

#### 注意事項

- 1) 関数の最大化、最小化、イコール 0 と置いた時の方程式の解を見つける際には、目的関数は `&` の印をつけてインプットとして呼び出す。
- 2) ローカル宣言では、目的関数には `:proc` とその属性を `procedure` であることを明記する。上の場合、`:fn` でもかまわない。

ただし、上の場合、始点 0 終点 1 の間をそのステップ 0.001 までの精度しか見ていない。この始点と終点の間に解（その関数の絶対値が 0 付近になる点）がなければこの計算は正しいとは言えない。その場合には、ファジー不等号の組込み関数 `fne` を用いて

```
_fcmptol=1e-2;
if fne(f(x0),0);
 errorlog "ERROR: Cannot find a solution.";
retp(".");
```

**endif;**

などというプログラムを解を見つける procedure 内のリターンの前に置いてやると、その解がはずれている場合には、引用符内のエラーログを出すようにできる。上では、0.001 刻みの Grid 上の点だけを見たわけで、組込み関数(a,b)という a と b が違う場合には 1 を、そうでない場合には 0 を  $1e-12$  の精度で返すものを使えばよいのだが、それでも精度は厳密すぎるので、刻みの 1 桁落ち程度の  $1e-2$  を ( すなわち 0.01 までは確実に a の部分の x0 と b の部分に入る 0 とは違うかどうか比べる ) その関数のグローバル設定である **\_fcmptol** に設定してやればうまくいくはずだ。この場合関数の値は、0 までの距離を比べているので  $f(x0)$  でも  $\text{abs}(f(x0))$  でも同じことである。同様に、上のプログラムは絶対値の組込み関数を外すことによって、その区間での最小値を求める procedure になる。また、最小値を与えるインデックスを求める minindc を最大値のもの maxindc に変更すれば、その区間の最大値を求める procedure に簡単にできる。一応、上では、始点を 0、終点を 0.9999 とすれば 0.0001 の 1 万点の高さを Light 版の制限内で比べられる。それ以上を比べるには、正式版を大容量メモリのもとで動作させるか、または、繰り返し計算をさせて、徐々に絞り込んでいくプログラムにすると方程式の解または最適点を求められる。少なくとも、2 値を求めるケースまでは確実にこの方法で求められる。

### 練習問題

【問 1】複数の procedure の扱い方のその 1 のところで、2 つのサイコロを 100 回振りその目の合計が 10「以上」になる確率を求めたい。そのプログラムを改造し答えを求めてみよう。

【問 2】最後の関数自体を別の procedure で最適化する方法で  $f(x) = -x^2 + 0.5$  について 0 と 1 の間の  $f(x) = 0$  とする時の解を求めてみよう。